Creating (PK-) concentration - event graphs

Berber Snoeijer, Biometric Support, Leiden, The Netherlands

ABSTRACT
Adverse events can be caused by high concentrations of study medication or related pharmacodynamic concentrations. To visualize the direct relation between the events and the drug concentration we created concentration profile-event graphs. In these graphs, the individual subject concentration-time profiles are presented together with bars depicting the actual start, duration and end of the adverse event(s). Profiles are created by treatment and kind of event showing bars with matching colors to the subject profiles. Additionally we also created profiles by subject showing bars with different colors for each kind of adverse event. We created these graphs in SAS® 8.2 and SAS® 9.1. Furthermore, we used SAS/BASE and SAS/GRAPH.
In this paper we present the key procedures to create these graphs based on a concentration-time table and a matching adverse event table.

INTRODUCTION
The safety and pharmacokinetics of a new drug are often tested together in the same study. During each study, the physician will mark whether a relation exists between the study drug and the adverse events (safety) that occurred during the study. At the end, summaries are made of these events by relationships to the study drug. However, from these summaries it is not clear whether a higher concentration results in a higher risk of the event occurring. To visualise that relation, it could be very helpful to create graphs in which the concentration-time curve is presented together with bars indicating the start and end time of each adverse event. In this paper we will present a method for preparing these graphs by subject and by treatment group.

CREATING INDIVIDUAL CONCENTRATION-EVENT GRAPHS
The individual concentration-event graph shows the concentration-time profile together with all the adverse events of interest that occurred for the specific subject. These graphs can be used to show what happened at a certain concentration level on an individual basis.

DATASETS USED
To create concentration-event graphs two datasets are needed: one for the concentration-time data and one for the adverse event data. In the concentration-time dataset we need at least variables for subject number (ptno), treatment number (trtno), treatment description (trtdesc), relative protocol time (time) and of course the concentrations (conc). In the adverse event dataset we need the variables for the subject number (ptno) and treatment number (trtno) as well to match both datasets. In addition, variables for adverse event description (aepref) and start and stop date/time relative to dosing are needed. Where the stop date is after the last time point presented in the graph we stored it in a new variable called ongoing. Each AE description will get a unique AE number (prefno) which is used for the annotation. You can choose to only present related adverse events or adverse events of interest. Other adverse events should not be included in the dataset.

ANNOTATE DATASETS
To present bars for each adverse event in the adverse event dataset, an annotate dataset has to be created. The code we use to do that is presented below. A new adverse event number (aeno) is added to the dataset which indicates the rank of the adverse event for the specific subject and counts the number of events for that subject. If the same event is reported more than once for a subject it will be given the same number. In that way the events with the same description and number are presented on the same vertical location in the graphs. Each subject that is included in the database gets a unique graph number. Furthermore, for each adverse event that is presented you can choose a specific filling style and specific colors for the corresponding bar. The styles and
Data chosen for a specific event type are preserved throughout all graphs regardless of whether only one or more events are reported for a specific subject.

```sas
DATA anno_indiv;
  SET selAEs;
  BY trt ptno prefno;
  RETAIN aeno grno prevae 0 ;
  LENGTH text $80. color $7.;
  /* getting unique number for each AE kind per subject */
  IF FIRST.ptno THEN DO; aeno=1; prevae=myAE; END;
  ELSE IF myAE NE prevae THEN DO; aeno = aeno+1; prevae=myAE; END;
  /* get new unique number per subject for graph no (grno) */
  IF FIRST.ptno THEN grno=grno+1;
  xsys='2';
  ysys='1';
  function="MOVE   ";
  /* give different filling styles to different kind of events */
  IF myae IN (1,3) THEN style="L2       "; ELSE style="SOLID";
  /* get start location of bars */
  x=start;
  y=(aeno-1)*3;
  /* add colors to each kind of ae */
  /* Other Valid colors: Gray, PINK, BLACK, BROWN, MAGENTA ORANGE */
  IF myAE=1 THEN COLOR='GREEN';
  ELSE IF myAE=2 THEN COLOR='MAGENTA';
  ELSE IF myAE=3 THEN COLOR='RED';
  OUTPUT;
  /* get location of end of the bar */
  function='BAR';
  x=stop;
  y=y+2;
  OUTPUT;
  END;
  CALL SYMPUT ('Gr_ind',grno);
RUN;
```

We have added code to visualize that events were ongoing after the x-axis ended and to present the time that they actually stopped. An arrow is added at the end of the bar and the stop time is given at the same location. Using call symput we stored the total number of graphs to be created in a macro variable. This macro is used in the graphs macro.

```sas
/* code in case of stop after end of graph */
IF ongoing>. THEN DO;
  function='LABEL';
  size=1;
  style='Arial';
  POSITION='6';
  COLOR='BLACK';
  TEXT='->';
  OUTPUT;
  y=y+2;
  x=22.2;
  TEXT='until '||TRIM(LEFT(PUT(ROUND(ongoing,.1),6.1)))||' h';
  OUTPUT;
END;
CALL SYMPUT ('Gr_ind',grno);
RUN;
```
More annotation was needed to add a legend for the specific bars of each adverse event. An example of the resulting graph is presented below.

**CREATING THE INDIVIDUAL GRAPHS**

To create the actual graphs we use a macro that loops for each subject that has reported an adverse event. First a subset of the created annotate dataset will be specified in which only the annotation of the specific subject is presented. Since the new graph numbers are not in the concentration-time dataset, we also need the subject number. Using call symput we get that from the annotate dataset and store it in a corresponding macro variable. The concentration time data is then plotted using the created annotate dataset.

```
%MACRO plotind; /* macro to create individual plots with selected AEs */
TITLE1 "Subject #BYVAL(ptno)";
TITLE2 "#BYVAL(trtdesc)" ;

%DO i=1 %TO &gr_ind;
DATA anno;
SET anno_indiv;
WHERE grno=&i;
CALL SYMPUT('ptno',PUT(ptno,8.));
RUN;
%PUT subject: &ptno;
PROC GPLOT DATA=conc UNIFORM NOCACHE ANNOTATE=anno;
BY ptno trtdesc;
WHERE ptno=&ptno;
PLOT conc*time / NOLEGEND VAXIS=axis1 HAXIS=axis2 
caxis=black ct=black c=black NOFRAME;
RUN;
QUIT;
%END;
%MEND;
```

**CREATING COMBINED CONCENTRATION-EVENT GRAPHS BY EVENT TYPE**

The combined concentration-event graph shows all concentration-time profiles for each treatment in one graph. Such a graph is created for each event of interest presenting bars for each subject that experiences the event. The
bars of each subject are given a different color that matches the color of the subject line. These graphs can be used to show the overall relation between the drug concentration and adverse events of interest.

DATASETS USED
The same datasets can be used as presented for the individual concentration event graphs.

ANNOTATE DATASETS
Although similar, the annotate dataset created for the combined graphs is different from that of the individual graphs. The code used to create this dataset is presented below. As for the individual annotate dataset, each graph that has to be created will be given a unique graph number (grno). For each kind of bar to be created a specific aeno is given. In this case different bars have to be created for different subjects. Thus the aeno is now related to the subject instead of to the adverse event.

DATA anno_comb;
   SET selAEs;
   BY prefno trtno ptno;
   RETAIN aeno grno prevsub 0 ;
   LENGTH text $80.  color $7.;
   /* get a unique number for each graph to be created */
   IF FIRST.trtno THEN grno=grno+1; /* grno = graph no: 1 graph for each AE kind and each treatment */
   /* get unique number for each bar */
   IF FIRST.trtno THEN DO; aeno=1; prevsub=ptno; END;
   ELSE IF ptno NE prevsub THEN DO; aeno = aeno+1; prevsub=ptno; END;
   xsy='2';
   ysys='1';
   function="MOVE   ";
   /* get start location of bars */
   x=start;
   y=(aeno-1)*3;
   /* add style and colors to each kind of ae */
   style="SOLID";
   IF AEno=1 THEN COLOR='RED';
   ELSE IF AEno=2 THEN COLOR='GREEN';
   ELSE IF AEno=3 THEN COLOR='MAGENTA';
   ELSE IF AEno=4 THEN COLOR='GRAY';
   ELSE IF AEno=5 THEN COLOR='BROWN'; /* maximum 5 subjects with an AE check:*/
   ELSE PUT 'WARNING: not enough colors defined';
   /* Other Valid colors: PINK, MAGENTA ORANGE */
   OUTPUT;
   /* get location of end of the bar */
   function='BAR';
   x=stop;
   y=y+2;
   OUTPUT;
   CALL SYMPUT ('Gr_comb',PUT(grno,8.));
RUN;

We use the same code as for the individual graphs to visualize that events were ongoing after the x-axis stopped (see individual graphs). Using call symput we stored the total number of graphs to be created in a macro variable which is used for creating the graphs. Also in this case, more annotation is needed to add a legend for the description of the specific bars for each subject.

CREATING THE COMBINED GRAPHS
To create the actual graphs we use a macro that loops for each treatment and event for which at least one adverse event was reported. First a subset of the annotate dataset will be created in which only the annotation of the specific subject is presented. Since the new graph numbers are not in the concentration-time dataset, we also need the treatment number and adverse event description. Using call symput we get that from the annotate dataset and store it in a corresponding macro variables. A subset of the concentration dataset has to be created as well because for each treatment you want to use a different annotation. The adverse event description has to be added to this concentration subset as we would like to present this description in the title.

```sas
%DO i=1 %TO &gr_comb;
DATA anno;
  SET anno_comb;
  WHERE grno=&i;
  CALL SYMPUT('trt',PUT(trtno,3.));
  CALL SYMPUT('aetxt',TRIM(aepref));
RUN;

DATA pc2;
  SET conc;
  WHERE trtno=&trt;
  aetxt="&aetxt";
RUN;

Futhermore, we need to match the colors of the bars with the colors of the lines. To do that we create datasets with all subject numbers in the treatment groups and add the corresponding color in case it was present in the annotate dataset. Thereafter, the corresponding color was stored in a macro variable. In case the corresponding adverse event was not reported for the specific subject the color was set to black. Finally, the graphs are created using the created subset of concentration-time data and of the annotate dataset.

```sas
DATA _NULL_;  
SET colors;  
/* if not in anno dataset then color=black */  
IF color='' THEN color='BLACK';  
CALL SYMPUT('col_sub'||LEFT(PUT(_N_,3.)),color);  
CALL SYMPUT('totsubs',PUT(_N_,3.));  
RUN;  
/* add colors to lines for the specific subjects */  
%DO k = 1 %TO &totsubs;  
SYMBOL&k   VALUE=&&value&k   INTERPOL=JOIN LINE=1  HEIGHT=1.1 WIDTH=1  
COLOR=&&col_sub&k REPEAT=1;  
%END;  
PROC GPLOT DATA=pc2 UNIFORM NOCACHE ANNOTATE=anno;  
BY aetxt trtdesc;  
PLOT conc*time=ptno / LEGEND=legend1 VAXIS=axis1 HAXIS=axis2  
caxis=black ct=black c=black NOFRAME;  
RUN;  
QUIT;  
%END;  
%MEND;  
```
CONCLUSION
In this paper a simple and efficient method is presented to visualize the relation between a concentration profile and any occurring event. Such profiles can be used to depict the relation between drug concentrations and adverse events but also for other concentration and event data. For example, the relation between efficacy data or a relation between PK and PD can be presented in the same manner.

CONTACT INFORMATION
Your comments and questions are valued and encouraged. Contact the author at:
Berber Snoeijer
Biometric Support
Einsteinweg 5C
2333 CC Leiden
Work Phone: +31 71 5721828
Fax: + 31 71 5765040
Email: bsnoeijer@biometricsupport.nl
Web: www.biometricsupport.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.